* Table A14 Impact of number of years of insurance payout on household outcomes
* intersection of sample used to study investments and agricultural outcomes and sample used to study household outcomes
* Last Updated, Niriksha Shetty 03/31/206

clear
clear mata
clear matrix
set maxvar 20000
set more off

pwd
cap cd c(`pwd')

estimates clear

use "../dta/reduced-form-master.dta", clear

* keeping intersection of sample
keep if ex_post_balp==1
keep if bal_panel3==1

* individual marketing treatments
local uniqinst "discount_2007 groupT muslimT hinduT video ppayT vframeT pframeT assigned_risk_ws_2008 discount_2008 rebate_50percentoff rebate2_1free rebate3_1free flyer_hyv bdmperc_2009 fourbdmperc_2009 disc4game_2009 bdmperc_2010 fourbdmperc_2010 disc4game_2010 assigned_risk_ws_2010 assigned_video_test assigned_drought_flyer assigned_subsidies_flyer assigned_loan bdmperc_2011 fourbdmperc_2011 disc4game_2011 bdmperc_2012 fourbdmperc_2012 disc4game_2012 bdmperc_2013 fourbdmperc_2013 disc4game_2013"  /* sewaT peerT assigned_risk_ws_2011 assigned_risk_ws_2012 assigned_risk_ws_2013 mrkt_allnegative mrkt_poslang mrkt_posimg*/

foreach var of varlist `uniqinst' {
	replace `var' = 0 if `var'==.
	replace `var' = 0 if wave==0
	}

foreach var of varlist `uniqinst' {
	replace `var' = 0 if treat_year==0
}

* Actual payout amount
replace payout_amt=payout_amt*1000


local table = 1
local rastring "replace"
local outcomevars "w_real_cons w_real_food_exp w_real_events_exp w_real_gifts_out w_real_nonfood_exp w_real_durables_exp w_real_income w_real_gifts_in w_real_sav w_real_lent w_real_borrow"
foreach var of local outcomevars {
replace `var' = 0 if `var' == .
}

foreach ovar of varlist `outcomevars' {
local col=1
	di "`ovar'"
	di "`table'"

	/* Village IV (No Fixed Effects) */
	ivreg2 `ovar' (payout_amt=treat_year) year2*  firstt_2007 lost_w3 resurveyed if c_b2009!=1, r cluster(villageno) 
	estimates store cT`table'C`col++'
	
	/* Village IV(FE/RE) */
	xtivreg2 `ovar' (payout_amt=treat_year) year2*  firstt_2007 lost_w3 resurveyed if c_b2009!=1, fe cluster(villageno)
	estimates store cT`table'C`col++'
	
	/* IND IV (No Fixed Effects) */
	ivreg2 `ovar' (payout_amt=treat_year `uniqinst') year2*  firstt_2007 lost_w3 resurveyed firstt_2009 , r cluster(villageno) 
	estimates store cT`table'C`col++'
	
	/* IND IV (Individual FE/RE) */
	xtivreg2 `ovar' (payout_amt=treat_year `uniqinst') year2*  firstt_2007 lost_w3 resurveyed firstt_2009, fe cluster(villageno)
	estimates store cT`table'C`col++'	

	qui estout cT`table'C* using "../output/a14.csv", ///
	cells(b(star fmt(3)) se(par(`"="("' `")""'))) stardetach  delimiter(",") ///
	starlevels(* .1 ** .05 *** .01) ///
	keep (payout_amt) prehead("`ovar'") ///
	mlabel("" "" "" "") ///
	`rastring' label stats(cdf N) 

local rastring "append"

}

* Well-being calculated at every 1000 rupees of payout amount

replace payout_amt = payout_amt/1000

local table = 1
local rastring "replace"
local outcomevars "food_sufficiency_child fladder_std outlook_std control_std"
foreach var of local outcomevars {
replace `var' = 0 if `var' == .
}

foreach ovar of varlist `outcomevars' {
local col=1
	di "`ovar'"
	di "`table'"

	/* Village IV (No Fixed Effects) */
	ivreg2 `ovar' (payout_amt=treat_year) year2*  firstt_2007 lost_w3 resurveyed if c_b2009!=1, r cluster(villageno) 
	estimates store cT`table'C`col++'
	
	/* Village IV(FE/RE) */
	xtivreg2 `ovar' (payout_amt=treat_year) year2*  firstt_2007 lost_w3 resurveyed if c_b2009!=1, fe cluster(villageno)
	estimates store cT`table'C`col++'
	
	/* IND IV (No Fixed Effects) */
	ivreg2 `ovar' (payout_amt=treat_year `uniqinst') year2*  firstt_2007 lost_w3 resurveyed firstt_2009 , r cluster(villageno) 
	estimates store cT`table'C`col++'
	
	/* IND IV (Individual FE/RE) */
	xtivreg2 `ovar' (payout_amt=treat_year `uniqinst') year2*  firstt_2007 lost_w3 resurveyed firstt_2009, fe cluster(villageno)
	estimates store cT`table'C`col++'	

	qui estout cT`table'C* using "../output/a14-b.csv", ///
	cells(b(star fmt(3)) se(par(`"="("' `")""'))) stardetach  delimiter(",") ///
	starlevels(* .1 ** .05 *** .01) ///
	keep (payout_amt) prehead("`ovar'") ///
	mlabel("" "" "" "") ///
	`rastring' label stats(cdf N) 

local rastring "append"

}
